

## COS341 Semester Project 2025 : The Students' Programming Language "SPL"

A fully operational compiler must be implemented as a group project with four students per group. The exact submission deadline will be announced via ClickUp. This work-sheet provides grammar and vocabulary for the compiler's front-end, such that you can start working. Further information concerning the compiler's back-end will be released later in separate work-sheets, whenever some relevant topics have been treated in the weekly lectures. You have plenty of "creative freedom" implement everything by means of your own methods, as you like, as long as the compiler emits correct target-code at the end of the project. And now: HAPPY WORKING :)

*General assumption:* No token contains any blank\_space. In other words: *if* the compiler encounters any blank\_space while "eating" the input string, the compiler assumes that another "new" token has its beginning "behind" the blank\_space. As "*blank\_space*" we also regard "*tabs*" or "*new\_line*". On the basis of these assumptions, this year's version of the "**Students' Programming Language**" (SPL) has the following **Context-Free Grammar** in which the **terminal** symbols are shaded with **green** colour for the sake of clarity:

---

SPL_PROG ::=	<b>glob</b> { VARIABLES } // comment: may have globals <b>proc</b> { PROCDEFS } // comment: without any return <b>func</b> { FUNCDEFS } // comment: return something <b>main</b> { MAINPROG }
VARIABLES ::=	// comment: nothing, nullable
VARIABLES ::=	VAR VARIABLES
VAR ::=	<u>user-defined-name</u> // comment: vocabulary: see below
NAME ::=	<u>user-defined-name</u>
PROCDEFS ::=	// comment: nothing, nullable
PROCDEFS ::=	PDEF PROCDEFS
PDEF ::=	NAME { PARAM } { BODY } // procedures return nothing
FDEF ::=	NAME { PARAM } { BODY ; <b>return</b> ATOM }
FUNCDEFS ::=	FDEF FUNCDEFS
FUNCDEFS ::=	// comment: nothing, nullable
BODY ::=	<b>local</b> { MAXTHREE } ALGO // may have local variables
PARAM ::=	MAXTHREE // We do not want to make the language SPL too complicated
MAXTHREE ::=	// comment: nothing, nullable
MAXTHREE ::=	VAR
MAXTHREE ::=	VAR VAR
MAXTHREE ::=	VAR VAR VAR // comment: for simplicity not more than three
MAINPROG ::=	<b>var</b> { VARIABLES } ALGO // may have local variables
ATOM ::=	VAR
ATOM ::=	<u>number</u> // comment: vocabulary: see below

ALGO	::=	INSTR <i>// in SPL we do not allow "empty" algorithms</i>
ALGO	::=	INSTR ; ALGO
INSTR	::=	halt
INSTR	::=	print OUTPUT
INSTR	::=	NAME ( INPUT ) <i>// comment: procedure call</i>
INSTR	::=	ASSIGN
INSTR	::=	LOOP
INSTR	::=	BRANCH
ASSIGN	::=	VAR = NAME ( INPUT ) <i>// comment: Function call</i>
ASSIGN	::=	VAR = TERM
LOOP	::=	while TERM { ALGO }
LOOP	::=	do { ALGO } until TERM
BRANCH	::=	if TERM { ALGO }
BRANCH	::=	if TERM { ALGO } else { ALGO }
OUTPUT	::=	ATOM
OUTPUT	::=	string <i>// comment: Vocabulary: see below</i>
INPUT	::=	<i>// comment: nothing, nullable</i>
INPUT	::=	ATOM
INPUT	::=	ATOM ATOM
INPUT	::=	ATOM ATOM ATOM <i>// comment: must match parameters!</i>
TERM	::=	ATOM
TERM	::=	( UNOP TERM )
TERM	::=	( TERM BINOP TERM )
UNOP	::=	neg
UNOP	::=	not
BINOP	::=	eq
BINOP	::=	>
BINOP	::=	or
BINOP	::=	and
BINOP	::=	plus
BINOP	::=	minus
BINOP	::=	mult
BINOP	::=	div

---

## Rules of the Vocabulary

1. A user-defined-name may **not** be identical with any **green keyword** of the Grammar!
2. Under the condition that Rule 1 is not violated, a user-defined-name is described by the **regular expression** `[a...z][a...z]*[0...9]*`
3. A constant number is has the **regular expression** `( 0 | [1...9][0...9]* )`
4. string : any sequence of digits or letters between "quotationmarks" : max. length 15!